One-Dimensional Analysis of Lateral Shearing Interferograms

James C. Wyant
Optical Sciences Center
University of Arizona

Introduction

Although techniques have been developed for obtaining the entire wavefront from two lateral shear interferograms, it is much easier, and often sufficient, to obtain the wavefront profile for a single scan across an interferogram. If the shear is sufficiently small a lateral shear interferogram gives the derivative of the wavefront in the direction of shear. For small shears the wavefront difference function can be fit to a polynomial and this polynomial can be integrated to obtain the wavefront. However, as the shear becomes larger it is no longer valid to assume the wavefront difference function is equal to the derivative. In these notes we illustrate an analysis method that is valid for both large and small lateral shear. The approach is to least-squares fit the wavefront difference function to a polynomial and then set this polynomial equal to the finite difference wavefront difference function and solve for the polynomial coefficients describing the wavefront in terms of the polynomial coefficients describing the wavefront difference function.

Interferogram data

Figure 1 shows the lateral shear interferogram being analyzed. A single scan line, parallel to the direction of shear, is drawn across the interferogram. The list "shearData" gives the x-coordinates of the intersection of the fringes with the scan line as well as the order number of the fringes. x_1, the left-hand edge coordinate of the lateral shear interferogram is measured to be 7.5 and x_2, the right-hand edge coordinate, is measured to be 51.5. The amount of shear is measured to be 7.5. We arbitrarily picked the order number of the fringes to initially increase in going from left to right and the first fringe was given an order number of 1. Changing the direction of the increase in fringe order number changes the sign of the departure. By simply looking at the interferogram it is impossible to determine the sign of the departure. The sign of the departure is determined during the photographing of the interferogram by use of conventional interferometric techniques such as changing the focus of the test setup. Changing the order number of the first fringe results in adding a constant to all the fringe order numbers which would change the slope of the calculated wavefront. Since the initial fringe order number is not known, the wavefront slope is not known, but the slope variation is known. The effect of picking an incorrect initial fringe order number is eliminated by adjusting the final data to give a zero overall slope.



Figure 1. Lateral shear interferogram showing central scan line parallel to the direction of shear.

List giving x fringe position and fringe order number.

```
shearData = {{8, 1}, {9, 2}, {10, 3}, {11.5, 4}, {13.5, 5}, {18, 6}, {23.5, 5}, {27, 4},
   {29.5, 3}, {32.5, 2}, {36, 1}, {41.5, 0}, {45.5, 1}, {47.5, 2}, {49, 3}, {50, 4}, {51, 5}};
```

Lateral Shear Interferogram Data

```
TableForm[shearData,
 TableHeadings -> {None, {"x position", "Fringe Order Number"}}, TableAlignments -> Center]
```

x position	Fringe Order Number
8	1
9	2
10	3
11.5	4
13.5	5
18	6
23.5	5
27	4
29.5	3
32.5	2
36	1
41.5	0
45.5	1
47.5	2
49	3
50	4
51	5

It is convenient to normalize the pupil x-coordinates to go from -1 to +1. Since in the original coordinates the pupil diameter was 51.5 all pupil coordinates must be divided by 51.5/2 and 1 must be subtracted from the result. The shear in the original coordinates was 7.5 and in the new coordinates is 7.5/(51.5/2) or the shear, Δ, is given by

```
shear = 7.5 / (51.5 / 2)
```

```
0.291262
```

To make the system symmetrical we will shift one of the two interfering beams to the left by shear/2 and the other interfering beam to the right by shear/2. Thus, in the new coordinates we must subtract shear/2 from each x-coordinate. The new x-coordinates of the fringes are given by

```
normalization = 51.5 / 2;
```

$$\text{Do}\left[\text{shearData}[[i, 1]] = \frac{\text{shearData}[[i, 1]]}{\text{normalization}} - 1 - \frac{\text{shear}}{2}, \{i, 1, \text{Length}[\text{shearData}]\}\right];$$

Lateral Shear Interferogram Data after x - coordinate Normalization

```
TableForm[Chop[shearData],
 TableHeadings -> {None, {"x position", "Fringe Order Number"}}, TableAlignments -> Center]
```

x position	Fringe Order Number
-0.834951	1
-0.796117	2
-0.757282	3
-0.699029	4
-0.621359	5
-0.446602	6
-0.23301	5
-0.0970874	4
0	3
0.116505	2
0.252427	1
0.466019	0
0.621359	1
0.699029	2
0.757282	3
0.796117	4
0.834951	5

Least Squares Polynomial Approach

One technique for analyzing a scan across a lateral shear interferogram first involves a least squares fit of the wave-front difference function data shown in Table 2 to a polynomial such as

$$\text{wdf[x\_, nMax\_] := } \sum_{n=0}^{nMax-1} b[n]\, x^n$$

For sufficiently small shears, the polynomial can be integrated to give the wavefront profile; however, for large shears, correction terms must be applied. If, in one dimension, the wavefront is written as

$$\Delta\text{w[x\_, nMax\_] := } \sum_{n=1}^{nMax} a[n]\, x^n$$

the wavefront difference function, v, can be written as

$$\text{v[x\_, nMax\_] := } \sum_{n=1}^{nMax} a[n] \left(\left(x + \frac{\Delta}{2}\right)^n - \left(x - \frac{\Delta}{2}\right)^n\right)$$

Rather than integrating wdf[x, nMax] to find Δw[x, nMax], a better approach is to set v[x, nMax] equal to wdf[x, nMax] and solve for the a[n]'s in terms of the b[n]'s. One approach for performing this calculation is shown below.

Solving for a[n]'s in terms of the b[n]'s

We will first find the difference between v[x, nMax] and wdf[x, nMax]. For our example we will let nMax = 4. The expression below finds the difference and groups the coefficients for each power of x from x^1 to x^{nMax}.

```
nMax = 4;
ans = CoefficientList[v[x, nMax] - wdf[x, nMax], x];
```

Since the difference is equal to zero for all values of x, we can set each coefficient equal to zero and solve for the a's.

```
Do[a[i] = (a[i] /. Flatten[Solve[ans[[i]] == 0, a[i]]]), {i, 1, nMax}]
```

The resulting equations for a[n] are

```
Print["Spherical  a[4]     ", Apart[a[4]]];
Print["Coma       a[3]     ", Apart[a[3]]];
Print["Defocus    a[2]     ", Apart[a[2]]]; Print["Tilt        a[1]     ", Apart[a[1]]];
```

Spherical a[4] $\frac{b[3]}{4\,\Delta}$

Coma a[3] $\frac{b[2]}{3\,\Delta}$

Defocus a[2] $\frac{b[1]}{2\,\Delta} - \frac{1}{8}\,\Delta\, b[3]$

Tilt a[1] $\frac{b[0]}{\Delta} - \frac{1}{12}\,\Delta\, b[2]$

Simple integration would have given only the first term in each expression above. Using the fact that a lateral shearing interferometer involves a finite-difference, rather than a derivative, makes it possible to obtain better results when the shear is not extremely small.

Analyzing data given in Table 2

The shearData can be fit to a polynomial as shown below.

```
Δ = shear;

Do[b[i - 1] = 0, {i, 1, nMax}];
fitShearData = Fit[shearData, {1, x, x^2, x^3}, x]
```

$3.05051 - 10.0708\,x - 0.0916631\,x^2 + 17.7765\,x^3$

```
b[0] = 3.05051; b[1] = -10.0708; b[2] = -0.0916631; b[3] = 17.7765;
```

■ Print wavefront difference function

These coefficients can be plugged into the expression for the wavefront difference function, wdf[x, nMax] and the results plotted as shown below.

```
Plot[wdf[x, nMax], {x, -1, 1}, Evaluate[plot2doptions]]
```



■ Print wavefront coefficients

The wavefront coefficients can be printed out. Since the wavefront difference function was calculated only through order 3, the wavefront is calculated through order 4.

```
Print["Spherical  a[4]    ", Apart[a[4]]];
Print["Coma       a[3]    ", Apart[a[3]]];
Print["Defocus    a[2]    ", Apart[a[2]]]; Print["Tilt       a[1]    ", Apart[a[1]]];
```

```
Spherical  a[4]    15.2582

Coma       a[3]    -0.104903

Defocus    a[2]    -17.9354

Tilt       a[1]    10.4756
```

■ Plot wavefront

The following shows a plot of the wavefront.

```
Plot[Δw[x, nMax], {x, -1, 1}, Evaluate[plot2doptions]]
```



Removing tilt we get

```
Plot[Δw[x, nMax] - Δw[x, 1], {x, -1, 1}, Evaluate[plot2doptions]]
```



Plotting only the 4th order coefficient yields

```
Plot[Δw[x, 4] - Δw[x, 3], {x, -1, 1}, Evaluate[plot2doptions]]
```